******************************************************************************** .

* HMWK2 .

* PURPOSE:  Run USP 510 HmWk2 exercises with SPSS MIXED for comparison with HLM5 .

* HISTORY:  Dennis Deck 2/13/04 .

* NOTES: Neighborhood drug crime

            Dependent:     NHRating (feelings about neighborhood)

                           Fear

            Person:        Sex

            Neighborhood:  Probs1  # of problems reported

                           Probs2  mean # reported

            Neighborhood ID: NH      .        

******************************************************************************** .

***                          Data Definition                                 *** .

******************************************************************************** .

TITLE 'USP 510 Multilevel Analysis - Homework 2' .

SUBTITLE 'Prepare analysis file with level 1 & 2 predictors'  .

* Aggregate level 2 file .

GET File= 'c:\hlm\hw2\aba_2.sav' /Map .

AGGREGATE OutFile= 'c:\hlm\hw2\aba_2a.sav' /PRESORTED /Break= NH

  /Probs2=FIRST(Probs2)  .

* Load level 1 file .

GET File= 'c:\hlm\hw2\aba_1.sav' /Map .

* Lookup level 2 variables .

MATCH FILES 

  /File=* 

  /Table= 'c:\hlm\hw2\aba_2a.sav'

  /BY=NH /Map .

/* Grand mean centered variables */ .

* compute and save means .

COMPUTE N = 1 .

AGGREGATE OutFile= 'c:\hlm\hw2\Means.sav' /PRESORTED /Break= N

  /ProbsM=MEAN(Probs1)  

  /FearM =MEAN(Fear) .

* lookup means .

MATCH FILES 

  File=*

 /Table='c:\hlm\hw2\Means.sav' 

 /By N .

* calculate grand mean centered variables .

COMPUTE Probs1C = Probs1 - ProbsM .

COMPUTE FearC = Fear - FearM .

LIST NH NHRating Fear Sex Probs1 Probs2 Probs1C FearC /Cases= 20 .

******************************************************************************** .

***                      Multilevel Analysis                                 *** .

******************************************************************************** .

SUBTITLE ' 1) Empty Model (Unconditional Means)' . 

MIXED NHrating 

  /Method= REML

  /Print= SOLUTION TESTCOV 

  /Fixed=  | SSTYPE(3)

  /Random= INTERCEPT | SUBJECT(NH) COVTYPE(un) .

SUBTITLE ' 2) add Probs1 ' . 

MIXED NHRating WITH Probs1

  /Method= REML

  /Print= SOLUTION TESTCOV 

  /Fixed= Probs1 | SSTYPE(3)

  /Random= INTERCEPT Probs1 | SUBJECT(NH) COVTYPE(un) .

SUBTITLE ' 3) rerun with Probs1C (grand mean centered)' .

MIXED NHRating WITH Probs1C

  /Method= REML

  /Print= SOLUTION TESTCOV 

  /Fixed= Probs1C | SSTYPE(3)

  /Random= INTERCEPT Probs1c | SUBJECT(NH) COVTYPE(un) .

SUBTITLE ' 4) rerun w/FearC (gr mean cntrd), Probs2, & interact' .

MIXED NHRating WITH FearC Probs2

  /Method= REML

  /Print= SOLUTION TESTCOV CORB

  /Fixed= FearC Probs2 FearC*Probs2 | SSTYPE(3)

  /Random= INTERCEPT FearC Probs2 | SUBJECT(NH) COVTYPE(un) .

SUBTITLE ' 6a) Probs1C and Probs2 ' . 

MIXED NHRating WITH Probs1C Probs2

  /Method= REML

  /Print= SOLUTION TESTCOV CORB

  /Fixed= Probs1C Probs2 | SSTYPE(3)

  /Random= INTERCEPT Probs1C Probs2 | SUBJECT(NH) COVTYPE(un) .

SUBTITLE ' 6b) Probs1C and Probs2, constrain Probs2' . 

MIXED NHRating WITH Probs1C Probs2

  /Method= REML

  /Print= SOLUTION TESTCOV CORB

  /Fixed= Probs1C Probs2 | SSTYPE(3)

  /Random= INTERCEPT Probs1c | SUBJECT(NH) COVTYPE(un) .

* END .

